



DECUS

PROGRAM LIBRARY

DECUS NO.	8-488
TITLE	NEWPAGE
AUTHOR	Geoffrey Chase
COMPANY	Portsmouth Abbey School Portsmouth, Rhode Island
DATE	September 20, 1971
SOURCE LANGUAGE	PAL III

DECEMBER



1900

Date	Description	Amount
12/1	To Balance	100.00
12/5	By Cash	50.00
12/10	By Cash	25.00
12/15	By Cash	75.00
12/20	By Cash	100.00
12/25	By Cash	150.00
12/30	By Cash	200.00
12/31	By Cash	250.00

"NEWPAGE"

A program to readdress binary tapes

Requirements: 4 K PDP-8 computer with teletype and paper tape reader/punch. Occupies locations 4000-4173 in any field. Starting address = 4000 (the program can be used to relocate itself).

Function: The program reads a binary tape on the low speed reader and punches out a relocated copy on the low speed punch. Address instructions are augmented by a number set in the switch register. This augment should be a multiple of octal 200 (one page); other values are reduced to the next lowest multiple of 200. The old checksum is ignored and a new one punched at the end of the output tape. The routine stops reading the source tape when it encounters the binary trailer at the end and halts, after punching checksum and trailer, at 4000. To restart, set SR for desired augment and press Continue.

Features: Addresses on page 0 of the source tape are not relocated, since page 0 usually contains pointers, constants, etc., which must be addressable from several different core pages. If a relocated address, not originally on page 0, lands on page 0 or page 37 (octal), the computer halts with the first two digits of the new address displayed in bits 6-11 of the AC. The user can then decide whether to proceed; if the new addresses are acceptable, he presses Continue on the console. A common cause of false alarms is the *200 punched by PAL III at the start of every binary tape, whatever the programmer's specified address may be. If one augments the addresses in such a tape by 7600 (= -200), the dummy address 200 will be relocated to 0000. The computer will halt, but the user, by pressing Continue, can proceed with the relocation.

Field Pseudo-Ops.: These are ignored completely by NEWPAGE, as are all codes above 200 (stray ASCII characters, for instance). Field instructions (CDF, etc.) are correctly copied.

High-speed Reader/Punch: A binary patch is supplied which will change input to the H.S.R. and output to the H.S.P. The leader and trailer are lengthened.

Copying RIM tapes: It would appear that this could be done by depositing 7000 in location 4027.

Use: Load address 4000 and start (PDP-8/E: clear + continue). The computer will halt immediately. Set the switch register for the number (octal) by which addresses are to be augmented. Check to see that the source tape is in the reader, the reader and punch turned on. Press Continue on the computer console.

*4000
 4000 7402 HLT /LOAD ADDRESS AUGMENT INTO S.R. ;
 /PRESS 'CONTINUE' ON CONSOLE.
 /AUGMENT WILL BE READ AS A MULTIPLE OF OCTAL 200,
 /I.E., AS AN INTEGER NO. OF PAGES.

4001	7604		LAS	
4002	7012		RTR	
4003	7012		RTR	
4004	7012		RTR	
4005	0361		AND K76	/A MULTIPLE OF 200
4006	3366		DCA AUGMNT	
4007	6032		KCC	
4010	3367		DCA FLAG	
4011	3373		DCA SUM	
4012	6046		TLS	
4013	4322		JMS LEADER	
4014	4245	PHASE1,	JMS TEST	
4015	3370		DCA STORE1	
4016	1367		TAD FLAG	
4017	7450		SNA	
4020	5214		JMP PHASE1	
4021	4245	PHASE2,	JMS TEST	
4022	3371		DCA STORE2	
4023	4245		JMS TEST	
4024	3372		DCA STORE3	
4025	1367		TAD FLAG	/REACHED TRAILER YET?
4026	7650		SNA CLA	
4027	5307		JMP EXIT	/YES
4030	1370		TAD STORE1	/NO; PUNCH DATA
4031	1373		TAD SUM	
4032	3373		DCA SUM	
4033	1370		TAD STORE1	
4034	4340		JMS PUNCH	
4035	1371		TAD STORE2	
4036	1373		TAD SUM	
4037	3373		DCA SUM	
4040	1371		TAD STORE2	
4041	4340		JMS PUNCH	
4042	1372		TAD STORE3	
4043	3370		DCA STORE1	
4044	5221		JMP PHASE2	
4045	0000	TEST,	Ø	
4046	4346		JMS READ	
4047	1360		TAD M100	
4050	7510		SPA	
4051	5262		JMP DATA	/CODE<100
4052	1360		TAD M100	
4053	7510		SPA	
4054	5264		JMP ADRS	/77<CODE<200
4055	7450		SNA	
4056	5260		JMP TRAIL	/CODE 200
4057	5246		JMP TEST+1	

4060	3367	TRAIL,	DCA FLAG	
4061	1363		TAD K100	
4062	1363	DATA,	TAD K100	
4063	5645		JMP I TEST	
4064	2367	ADRS,	ISZ FLAG	
4065	1363		TAD K100	
4066	1355		TAD M1	
4067	7550		SPA SNA	/SKIP IF ADDRESS IS /NOT ON PAGE 0
4070	5305		JMP ZADRS	
4071	7001		IAC	/RESTORE 1
4072	1366		TAD AUGMNT	/READDRESS
4073	0362		AND K77	
4074	1355		TAD M1	/LANDS ON PAGE 0?
4075	7550		SPA SNA	
4076	5332		JMP ERROR0	/YES
4077	7001		IAC	/NO; RESTORE 1
4100	1357	TEST37,	TAD M76	/LANDS ON LAST PAGE?
4101	7500		SMA	
4102	5335		JMP ERROR	/YES
4103	1361		TAD K76	/NO; RESTORE 76
4104	5262		JMP DATA	/RESTORE 100 & PUNCH
4105	7001	ZADRS,	IAC	/RESTORE 1
4106	5262		JMP DATA	/RESTORE 100 & PUNCH
4107	1373	EXIT,	TAD SUM	
4110	7012		RTR	
4111	7012		RTR	
4112	7012		RTR	
4113	0362		AND K77	
4114	4340		JMS PUNCH	
4115	1373		TAD SUM	
4116	0362		AND K77	
4117	4340		JMS PUNCH	
4120	4322		JMS LEADER	
4121	5200		JMP 4000	
4122	0000	LEADER,	0	
4123	1356		TAD M30	
4124	3370		DCA STORE1	
4125	1364		TAD K200	
4126	4340		JMS PUNCH	
4127	2370		ISZ STORE1	
4130	5325		JMP .-3	
4131	5722		JMP I LEADER	
4132	7001	ERROR0,	IAC	
4133	7402		HLT	/MISTAKE? IF NOT, CONTINUE
4134	5300		JMP TEST37	
4135	1361	ERROR,	TAD K76	
4136	7402		HLT	/MISTAKE?
4137	5262		JMP DATA	

```

4140 0000 PUNCH, 0
4141 6041 TSF
4142 5341 JMP --1
4143 6046 TLS
4144 7200 CLA
4145 5740 JMP I PUNCH
4146 0000 READ, 0
4147 7000 NOP /FOR H.S.R. PATCH
4150 6031 KSF
4151 5350 JMP --1
4152 7200 CLA
4153 6036 KRB
4154 5746 JMP I READ

```

```

4155 7777 M1, -1
4156 7750 M30, -30
4157 7702 M76, -76
4160 7700 M100, -100
4161 0076 K76, 76
4162 0077 K77, 77
4163 0100 K100, 100
4164 0200 K200, 200
4165 0207 K207, 207
4166 0000 AUGMNT, 0
4167 0000 FLAG, 0
4170 0000 STORE1, 0
4171 0000 STORE2, 0
4172 0000 STORE3, 0
4173 0000 SUM, 0

```

PAUSE

/THE COMPUTER HALTS IF AN AUGMENTED ADDRESS
 /LANDS ON PAGE 0 OR ON (OCTAL) PAGE 37. LIGHTS
 /6-11 OF THE ACCUMULATOR WILL SHOW THE FIRST TWO
 /DIGITS OF THE NEW ADDRESS. TO OVERRIDE THE HALT,
 /PRESS 'CONTINUE' ON THE CONSOLE. ADDRESSES ORIGI-
 /NALLY LOCATED ON PAGE 0 ARE LEFT UNCHANGED.

ADRS	4064
AUGMNT	4166
DATA	4062
ERROR	4135
ERROR0	4132
EXIT	4107
FLAG	4167
K100	4163
K200	4164
K207	4165
K76	4161
K77	4162
LEADER	4122
M1	4155
M100	4160
M30	4156
M76	4157
PHASE1	4014
PHASE2	4021
PUNCH	4140
READ	4146
STORE1	4170
STORE2	4171
STORE3	4172
SUM	4173
TEST	4045
TEST37	4100
TRAIL	4060
ZADRS	4105

/PATCHES TO 'NEWPAGE' FOR HIGH SPEED
/READER AND PUNCH.

		*4007	
4007	7000		NOP
		*4012	
4012	6026		PLS
		*4141	
4141	6021		PSF
4142	5341		JMP .-1
4143	6026		PLS
		*4147	
4147	6014		RFC
4150	6011		RSF
4151	5350		JMP .-1
4152	7200		CLA
4153	6012		RRB

		*4123	
4123	1360		TAD M100

M100=4160

M100	4160
------	------

